


```
> restart;
```

The House heating model

We have a house with two rooms A and B and B has a furnace F in it
A(t) is the temperature in room A and
B(t) is the temperature in room B and
Am is the outside air temperature.

The following figure was created in Maple: See Insert->Drawing



Let A(t) be the temp. in room A at time t and
B(t) be the temp in room B at time t.

```
> de1 := diff(A(t),t) = +k1*(B(t)-A(t))-k2*(A(t)-Am);
```

$$de1 := \frac{d}{dt} A(t) = k1 (B(t) - A(t)) - k2 (A(t) - Am)$$

```
> de2 := diff(B(t),t) = -k1*(B(t)-A(t))-k3*(B(t)-Am)+F;
```

$$de2 := \frac{d}{dt} B(t) = -k1 (B(t) - A(t)) - k3 (B(t) - Am) + F$$

```
> TempEquil := solve( map( rhs, {de1,de2} ), {A(t),B(t)} );
```

$$\text{TempEquil} := \left\{ A(t) = \frac{Am k1 k2 + Am k1 k3 + Am k2 k3 + Fk1}{k1 k2 + k1 k3 + k2 k3}, B(t) \right. \\ \left. = \frac{Am k1 k2 + Am k1 k3 + Am k2 k3 + Fk1 + Fk2}{k1 k2 + k1 k3 + k2 k3} \right\}$$

```
> f := x^2*y+x^2+y*x+x+2+y^2;
```

$$f := x^2 y + x^2 + x y + y^2 + x + 2$$

```
> collect(f,x);
```

$$(y+1)x^2 + (y+1)x + y^2 + 2$$

> collect(f,x,F);

$$F(y+1)x^2 + F(y+1)x + F(y^2 + 2)$$

> collect(TempEquil, F);

$$\left\{ A(t) = \frac{k_1 F}{k_1 k_2 + k_1 k_3 + k_2 k_3} + \frac{Am k_1 k_2 + Am k_1 k_3 + Am k_2 k_3}{k_1 k_2 + k_1 k_3 + k_2 k_3}, B(t) \right. \\ \left. = \frac{(k_1 + k_2) F}{k_1 k_2 + k_1 k_3 + k_2 k_3} + \frac{Am k_1 k_2 + Am k_1 k_3 + Am k_2 k_3}{k_1 k_2 + k_1 k_3 + k_2 k_3} \right\}$$

The following shows that the temperature in room A(t) and B(t) is in the form $Am + C \cdot F$ where C is a function of the parameters. This is a nice result.

> collect(TempEquil, F, simplify);

$$\left\{ A(t) = \frac{k_1 F}{k_1 k_2 + k_1 k_3 + k_2 k_3} + Am, B(t) = \frac{(k_1 + k_2) F}{k_1 k_2 + k_1 k_3 + k_2 k_3} + Am \right\}$$

> Am := 0;

k2 := k3;

k1 := 2*k2;

$$Am := 0$$

$$k2 := k3$$

$$k1 := 2 k3$$

> dsolve({de1,de2,A(0)=Am,B(0)=Am}, {A(t),B(t)});

$$\left\{ A(t) = \frac{1}{5} \frac{-\frac{5}{2} e^{-tk_3} F + \frac{1}{2} e^{-5tk_3} F + 2 F}{k_3}, B(t) = \frac{1}{5} \frac{-\frac{1}{2} e^{-5tk_3} F - \frac{5}{2} e^{-tk_3} F + 3 F}{k_3} \right\}$$

> F := 5;

k3 := 0.1;


$$F := 5$$

$$k3 := 0.1$$

> sol := dsolve({de1,de2,A(0)=Am,B(0)=Am}, {A(t),B(t)});

$$sol := \left\{ A(t) = 5 e^{-\frac{1}{2} t} - 25 e^{-\frac{1}{10} t} + 20, B(t) = -5 e^{-\frac{1}{2} t} - 25 e^{-\frac{1}{10} t} + 30 \right\}$$

> plot(map(rhs,sol), t=0..50);



> TempEquil;

$$\{A(t) = 20.00000000, B(t) = 30.00000000\}$$

> map(limit, sol, t=infinity);

$$\{\lim_{t \rightarrow \infty} A(t) = 20, \lim_{t \rightarrow \infty} B(t) = 30\}$$

> with(DEtools):

DEplot({de1,de2}, {A(t),B(t)}, t=0..40, A=0..50,B=0..50,
[[A(0)=0,B(0)=0],[A(0)=50,B(0)=30]], linecolor=blue, arrows=
medium);

